Név: Anonymous

1. dolgozat
Számítógéppel segített matematikai modellezés
"A" változat
2012. október 30, kedd

Oldd meg a következő feladatokat. Készíts szép notebook-ot, figyelj a korrekt strukturált megoldásokra, válaszolj a kérdésekre.

---

1. feladat (15 pont)

a) Készítsd el az alábbi táblázat elsô 50 sorát!

a j-edik sorban a j-edik természetes szám áll, utána a j-edik ikerprím(pár) elsô és második tagja, tehát pl. a 2. sor:

2   5   7

In[23]:=

```
T = Select[Table[Prime[n], {n, 250}], PrimeQ[# + 2] &]
```

Out[23]=

```
{3, 5, 11, 17, 29, 41, 59, 71, 101, 107, 137, 149, 179, 191, 197, 227, 239, 269, 281,
 311, 347, 419, 431, 461, 521, 569, 599, 617, 641, 659, 809, 821, 827, 857, 881, 1019,
 1031, 1049, 1061, 1091, 1151, 1229, 1277, 1289, 1301, 1319, 1427, 1451, 1481, 1487}
```

In[26]:=
```
TableForm[Table[{Style[n, Red], T⟦n⟧, T⟦n⟧ + 2}, {n, Length[T]}]]
```

Out[26]//TableForm=

1	3	5
2	5	7
3	11	13
4	17	19
5	29	31
6	41	43
7	59	61
8	71	73
9	101	103
10	107	109
11	137	139
12	149	151
13	179	181
14	191	193
15	197	199
16	227	229
17	239	241
18	269	271
19	281	283
20	311	313
21	347	349
22	419	421
23	431	433
24	461	463
25	521	523
26	569	571
27	599	601
28	617	619
29	641	643
30	659	661
31	809	811
32	821	823
33	827	829
34	857	859
35	881	883
36	1019	1021
37	1031	1033
38	1049	1051
39	1061	1063
40	1091	1093
41	1151	1153
42	1229	1231
43	1277	1279
44	1289	1291
45	1301	1303
46	1319	1321
47	1427	1429
48	1451	1453
49	1481	1483
50	1487	1489

b) Add meg a szupertökéletes számokat 100000-ig, azaz olyan n természetes számokat generálj, melyre $((\sigma(\sigma(n)) = 2n) \wedge (n<10^6))$. Pl. $(2*2=\sigma(\sigma(2))=\sigma(3)=4$, σ-osztók összege)

In[30]:=
```
Select[Table[n, {n, 10^5}], Plus @@ Divisors[Plus @@ Divisors[#]] == 2 # &]
```

Out[30]= {2, 4, 16, 64, 4096, 65536}

In[113]:=
```
Select[Table[n, {n, 10^5}], DivisorSigma[1, DivisorSigma[1, #]] == 2 # &]
```

Out[113]= {2, 4, 16, 64, 4096, 65536}

c) Add meg az $f(x) = x^2 e^{-x}$ fgv $f^{(k)}$ magasabbrendû deriváltjait táblázatban(0≤k≤10) . Adj ez alapján képletet $f^{(n)}(x)$-re! (n∈ℕ). Add meg az $f^{(k)}(0)$ sorozatot is (0≤k≤10).

In[1]:=
```
f[x_] := x^2 E^-x
```

In[4]:=
```
Factor[D[f[x], x]]
```

Out[4]= $-e^{-x}(-2+x)\,x$

In[6]:=
```
TableForm[Table[{k, Factor[D[f[x], {x, k}]]}, {k, 0, 10}]]
```

Out[6]//TableForm=

0	$e^{-x} x^2$
1	$-e^{-x}(-2+x)\,x$
2	$e^{-x}(2-4x+x^2)$
3	$-e^{-x}(6-6x+x^2)$
4	$e^{-x}(-6+x)(-2+x)$
5	$-e^{-x}(20-10x+x^2)$
6	$e^{-x}(30-12x+x^2)$
7	$-e^{-x}(42-14x+x^2)$
8	$e^{-x}(56-16x+x^2)$
9	$-e^{-x}(-12+x)(-6+x)$
10	$e^{-x}(90-20x+x^2)$

In[8]:=
```
Table[Factor[D[f[x], {x, k}]], {k, 0, 10}] /. x → 0
```

Out[8]= {0, 0, 2, -6, 12, -20, 30, -42, 56, -72, 90}

Válasz: $e^{-x}(-1)^k\,(x^2-(2k)\,x+k(k+1))$

---

2. feladat (10 pont)

Ábrázold a

$$h(x) = \frac{1}{1/2-e^{1/x}}$$

függvényt a [-10,10] intervallumon, add meg az értelemezési tartományt!
Add meg +∞-ben és a -∞-ben a limeszt!
Add meg a két véges kritikus pontban a féloldali limeszeket! Van-e limesz?
Hogyan kell az elsô ábrát módosítani, hogy a véges pontokban a féloldali limeszek vizuálisan is szemléltethetôk legyenek?
Add meg az inflexiós pontokat, színezz (új ábrán) a második derivált elôjele szerint!

In[32]:=
```
f[x_] = 1 / (1 / 2 - E^(1 / x));
```

In[33]:=
```
Solve[Denominator[f[x]] == 0, x]
```

Solve::ifun : Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information. ≫

Out[33]= $\left\{\left\{x \to -\frac{1}{\text{Log}[2]}\right\}\right\}$

In[34]:= `Plot[f[x], {x, -10, 10}]`

Out[34]=



In[41]:= `Limit[f[x], x → ∞]`

Out[41]= -2

In[42]:= `Limit[f[x], x → -∞]`

Out[42]= -2

In[35]:= `Limit[f[x], x → -1 / Log[2], Direction → -1]`

Out[35]= ∞

In[36]:= `Limit[f[x], x → -1 / Log[2], Direction → 1]`

Out[36]= $-\infty$

In[37]:= `Plot[f[x], {x, -2, 0}]`

Out[37]=



In[39]:= `Limit[f[x], x → 0, Direction → -1]`

Out[39]= 0

In[40]:= `Limit[f[x], x → 0, Direction → 1]`

Out[40]= 2

In[38]:=
```
Plot[f[x], {x, -1, 1}]
```

Out[38]=



In[53]:=
```
ip = {x, f[x]} /. NSolve[D[f[x], x, x] == 0, x, Reals]
```

Out[53]= {{0.606303, -0.212605}}

Csak egy inf. pont van x0~0.6

In[46]:=
```
Reduce[D[f[x], x, x] > 0, x, Reals]
```

Out[46]= $-\frac{1}{\text{Log}[2]} < x < 0 \,||\, x > \text{Root}\left[\left\{-1 - 2\,\#1 + e^{\frac{1}{\#1}}\,(-2 + 4\,\#1)\,\&,\ 0.60630266641497825512\right\}\right]$

In[47]:=
```
N[%]
```

Out[47]= -1.4427 < x < 0. || x > 0.606303

In[62]:=
```
Plot[f[x], {x, -5, 5}, PlotRange → {-6, 6},
 ColorFunction → Function[{x}, If[(D[f[z], z, z] /. z → x) > 0, Red, Blue]],
 ColorFunctionScaling → False, Epilog → {PointSize[.02], Green, Point[ip]}]
```

Out[62]=



---

3. feladat (15 pont)

A numerikus matematikában az $L(x) = \sum_{j=1}^{n} \left| \prod_{k=1,\,k\neq j}^{n} (x - x_k) \big/ (x_j - x_k) \right|$ polinomot az XL = $\{x_1, x_2\ \ldots,\ x_n\}$ alappontrendszerhez (elemek pár. kül.) tartozó Lebesgue polinomnak nevezik.

a) Add meg azt a függvényt, aminek bemenô paramétere az XL lista és elôállítja L-t.
b) Add meg LebF[{-1, 0, 1}]-et, ábrázold [-1, 1]-en.
c) Add meg LebF[{ -1, 0, 1}] maximumát [-1, 1]-en, jelöld be azokat a pontokat a b) -beli grafikonon, ahol ezt a maximumot felveszi a függvény!

Készíts interaktív ábrát melyben 2 csúszka segítségével ábrázolod a LebF[{a1,0,a3}] függvénycsaládot. ($a1,a3\neq 0$ $a1\neq a3$, $a1,a3\in[-1, 1]$)

Tippek: Sum, Product, Abs, Maximize, NMaximize, Manipulate

Itt egy tipikus példa:

```
LebF[{-1, -1/3, 1/3, 1}]
```

$$\frac{9}{16}\text{Abs}\left[(1-x)\left(-\frac{1}{3}+x\right)\left(\frac{1}{3}+x\right)\right]+\frac{27}{16}\text{Abs}\left[(-1+x)\left(-\frac{1}{3}+x\right)(1+x)\right]+\frac{27}{16}\text{Abs}\left[(-1+x)\left(\frac{1}{3}+x\right)(1+x)\right]+\frac{9}{16}\text{Abs}\left[\left(-\frac{1}{3}+x\right)\left(\frac{1}{3}+x\right)(1+x)\right]$$

In[63]:=
```
LebF[XL_List] := Sum[Abs[Product[(x - XL[[k]]) / (XL[[j]] - XL[[k]]), {k, 1, j - 1}]
    Product[(x - XL[[k]]) / (XL[[j]] - XL[[k]]), {k, j + 1, Length[XL]}]], {j, 1, Length[XL]}]
```

In[65]:=
```
LebF[{-1, -1/3, 1/3, 1}]
```

Out[65]=
$$\frac{9}{16}\text{Abs}\left[(1-x)\left(-\frac{1}{3}+x\right)\left(\frac{1}{3}+x\right)\right]+\frac{27}{16}\text{Abs}\left[(-1+x)\left(-\frac{1}{3}+x\right)(1+x)\right]+\frac{27}{16}\text{Abs}\left[(-1+x)\left(\frac{1}{3}+x\right)(1+x)\right]+\frac{9}{16}\text{Abs}\left[\left(-\frac{1}{3}+x\right)\left(\frac{1}{3}+x\right)(1+x)\right]$$

In[66]:=
```
LebF[{-1, 0, 1}]
```

Out[66]=
$$\frac{1}{2}\text{Abs}[(1-x)\,x]+\text{Abs}[(1-x)(1+x)]+\frac{1}{2}\text{Abs}[x\,(1+x)]$$

In[68]:=
```
Plot[LebF[{-1, 0, 1}], {x, -1, 1}]
```

Out[68]=


In[72]:=
```
Maximize[{LebF[{-1, 0, 1}], -1 ≤ x ≤ 1}, x]
```

Out[72]=
$$\left\{\frac{5}{4},\left\{x\to-\frac{1}{2}\right\}\right\}$$

In[73]:=
```
mlist = {{-1/2, 5/4}, {1/2, 5/4}};
```

In[76]:=
```
Plot[LebF[{-1, 0, 1}], {x, -1, 1}, Epilog → {Green, PointSize[.02], Point[mlist]}]
```

Out[76]=



In[78]:=
```
Manipulate[Plot[LebF[{a1, 0, a3}], {x, -1, 1}, PlotRange → {{-1.1, 1.1}, {-.1, 3}}],
 {a1, -1, 1, .1}, {a3, -1, 1, .1}]
```

Out[78]=



In[82]:=
```
Manipulate[Plot[If[a1 == 0 ∨ a3 == 0 ∨ a1 == a3, -1, LebF[{a1, 0, a3}]],
  {x, -1, 1}, PlotRange → {{-1.1, 1.1}, {-.1, 6}}], {a1, -1, 1, .1}, {a3, -1, 1, .1}]
```

Out[82]=



4. feladat (15 pont)

Egy számot “majdnem prím”-nek vagy félprímnek nevezük, ha pontosan 2 kül prímtényezôje van multiplicitással, azaz ha $z = p1^{a1}\ p2^{a2} .. pk^{ak}$, akkor $\sum_i a_i = 2$.

Add meg az elsô 50 ilyen elemet. Készíts interkatív ábrát is: a csúszka (n) menjen 1 és 50 között és adja meg a az n-edik félprímet és a prímfelbontását .

Készíts kísérletet a relatív gyakoriságuk vizsgálatára. Legyen $n = 10^k$, $k = 2, 3, 4, 5, 6$. Ábrázold a sorozatot.

Tipp: FactorInteger , ListPlot

In[95]:=
```
SPL = Select[Table[n, {n, 150}], Plus @@ FactorInteger[#][[All, 2]] == 2 &]
```

Out[95]= {4, 6, 9, 10, 14, 15, 21, 22, 25, 26, 33, 34, 35, 38, 39, 46, 49, 51, 55, 57, 58, 62, 65, 69, 74, 77, 82, 85, 86, 87, 91, 93, 94, 95, 106, 111, 115, 118, 119, 121, 122, 123, 129, 133, 134, 141, 142, 143, 145, 146}

In[96]:=
```
Length[%]
```

Out[96]= 50

In[89]:=
```
Manipulate[TableForm[{{Style[n, Blue], SPL[[n]], FactorInteger[SPL[[n]]]}}], {n, 1, 50, 1}]
```

Out[89]=



In[94]:=
```
Manipulate[TableForm[{{Style[n, Blue], SPL[[n]],
   If[(l = FactorInteger[SPL[[n]]]); Length[l] == 1, StringJoin[ToString[l[[1, 1]]], "^2"],
    StringJoin[ToString[l[[1, 1]]], "*", ToString[l[[2, 1]]]]]}}], {n, 1, 50, 1}]
```

Out[94]=



In[105]:=
```
T2 = N[Table[
   Length[Select[Table[j, {j, 10^k}], Plus @@ FactorInteger[#][[All, 2]] == 2 &]] / 10^k, {k, 2, 6}]]
```

Out[105]= {0.34, 0.299, 0.2625, 0.23378, 0.210035}

In[110]:=
```
ListPlot[Transpose[{Table[k, {k, 2, 6}], T2}], PlotRange → {{0, 7}, {0, 1 / 2}}]
```

Out[110]=